

Convergencia de sucesiones y series de funciones

El objetivo de esta práctica es usar el programa *Mathematica* para ilustrar mediante animaciones gráficas los conceptos de convergencia puntual y uniforme. Al mismo tiempo, aprenderemos algunos *comandos* y procedimientos de *Mathematica* que tienen interés por sí mismos.

Como sabes, el *campo de convergencia puntual* de una sucesión de funciones $\{f_n\}$ que suponemos definidas en un intervalo I , es el conjunto $C = \{x \in I : \{f_n(x)\} \text{ es convergente}\}$; y, supuesto que $C \neq \emptyset$, la *función límite puntual* de $\{f_n\}$ es la función $f: C \rightarrow \mathbb{R}$ dada por $f(x) = \lim\{f_n(x)\}$ para todo $x \in C$. Se dice también que la sucesión $\{f_n\}$ *converge puntualmente* en C .

Dado un intervalo $J \subseteq C$, se dice que $\{f_n\}$ *converge uniformemente en J* si para todo $\varepsilon > 0$ existe un número natural n_ε tal que para todo $n \geq n_\varepsilon$ se verifica que $\sup\{|f_n(x) - f(x)| : x \in J\} \leq \varepsilon$.

Ejemplo 1

Definamos las funciones $f_n: [0, 1] \rightarrow \mathbb{R}$ por $f_n(x) = x^n(1 - x^n)$. Vamos a representar juntas las gráficas de las primeras quince funciones de esta sucesión.

Recuerda que el comando `Plot[f, {x, xmin, xmax}]` dibuja la gráfica de f como función de x en el intervalo $[xmin, xmax]$; y `Plot[{f1, f2, ...}, {x, xmin, xmax}]` dibuja juntas las gráficas de las funciones f_1, f_2, \dots . Observa que $\{f_1, f_2, \dots\}$ es una *lista*. Una *lista* es simplemente cualquier objeto encerrado entre llaves $\{ \}$. *Mathematica* utiliza listas para fines muy diversos, como para definir vectores o matrices y en muchas otras situaciones.

Como escribir la lista con las primeras quince funciones de nuestra sucesión es bastante pesado, podemos pedirle a *Mathematica* que lo haga. Una forma usual de generar listas con *Mathematica* la proporciona el comando `Table[expr, {i, imax}]` que genera una lista con los valores de `expr` cuando i recorre los enteros de 1 a $imax$.

Debido a la forma en que `Plot` evalúa sus argumentos, asignando primero valores a la variable x y evaluando después las funciones en dichos valores, cuando la lista de funciones de `Plot` viene dada por un comando como, por ejemplo, `Table`, es preciso usar `Evaluate` para que *Mathematica* haga primero la lista de funciones y después las evalúe (pues de otra forma *Mathematica* trataría de producir una nueva tabla para cada valor de x lo que produce un error de salida).

Recordemos que `Plot` tiene muchas opciones, una de ellas es `PlotRange->{ymin, ymax}` que sirve para indicar el intervalo del eje de ordenadas donde queremos que se representen las funciones.

Ahora ya puedes entrar en *Mathematica* la siguiente orden:

```
In[1]:=
Plot[Evaluate[Table[x^i(1-x^i), {i, 15}]],
{x, 0, 1}, PlotRange->{0, 0.3}]
Out[1]=
-Graphics-
```

Es inmediato que $\lim\{f_n(x)\} = 0$, para todo $x \in [0, 1]$; pero al observar las gráficas te darás cuenta de que cuando x está *muy próximo* a 1 la sucesión $\{f_n(x)\}$ converge a 0 *muy lentamente*. Podemos hacernos idea de la rapidez con que $\{f_n(x)\}$ converge a 0 pidiéndole a *Mathematica* que haga una lista con los primeros n términos de $\{f_n(x)\}$, lista que después evaluaremos para distintos valores de x y de n .

```
In[2]:=
f[n_,x_]:=x^n(1-x^n)

In[3]:=
sucesion[n_,x_]:=Table[f[i,x],{i,n}]
```

Si ahora escribes `sucesion[1000,0.99]` obtendrás una lista con las primeras mil funciones de la sucesión evaluadas en $x = 0,99$. Observa que los valores en la lista van creciendo hasta llegar a $0,25$ y después decrecen hasta hacerse muy pequeños, de hecho el último valor es menor que 10^{-4} . Si ahora escribes `sucesion[1000,0.999]` verás que también los valores en la lista crecen hasta llegar a $0,25$ y después decrecen pero ya no se hacen tan pequeños. Podemos intentar calcular m tal que para $n \geq m$ tengamos que $f_n(0,999) < 10^{-4}$. Como, evidentemente, $f_n(x) \leq x^n$, es suficiente tomar m tal que $(0,999)^m < 10^{-4}$. Tomando logaritmos y despejando resulta que:

```
In[4]:=
m=Floor[Log[10^-4]/Log[0.999]]+1
Out[4]=
9206
```

Donde hemos usado la función parte entera que en *Mathematica* se escribe `Floor[x]` y da el mayor entero menor o igual que x . Puedes probar ahora con $x = 0,9999$ y obtendrás $m = 92099$. Ya ves que cuanto más próximo está x a 1 hay que tomar valores mayores de n para lograr que $f_n(x) < 10^{-4}$.

Lo anterior nos indica que $\{f_n\}$ no converge uniformemente en $[0,1]$. De hecho, observando las gráficas que obtuvimos antes, parece que el mayor valor que alcanzan las funciones de la sucesión en dicho intervalo es igual a $0,25$. Para comprobarlo podemos estudiar la derivada de f_n .

```
In[5]:=
Simplify[D[f[n,x],x]]
Out[5]=
-nx^(-1+n)(-1+2x^n)
```

La forma de la derivada permite deducir fácilmente que f_n es creciente en $[0, \frac{1}{\sqrt[n]{2}}]$ y decreciente en $[\frac{1}{\sqrt[n]{2}}, 1]$ por lo que $\max\{f_n(x) : x \in [0,1]\} = f_n(\frac{1}{\sqrt[n]{2}}) = 1/4$. Hemos probado así que $\{f_n\}$ no converge uniformemente en $[0,1]$.

El estudio anterior sugiere que puede haber convergencia uniforme en intervalos de la forma $[0, a]$ donde $0 < a < 1$. En efecto, dado $0 < a < 1$, tomemos $p \in \mathbb{N}$ tal que $a < \frac{1}{\sqrt[p]{2}}$. Entonces, para todo $n \geq p$ tenemos que $a \in [0, \frac{1}{\sqrt[p]{2}}] \subseteq [0, \frac{1}{\sqrt[n]{2}}]$ por lo que $\max\{f_n(x) : x \in [0, a]\} = f_n(a)$, y como $\lim\{f_n(a)\} = 0$ concluimos que $\{f_n\}$ converge uniformemente en $[0, a]$.

Modificando un poco la gráfica que hicimos al principio podemos obtener una animación con las gráficas de las primeras quince funciones. Basta para ello con hacer una tabla con las quince gráficas.

```

In[6]:=
Table[Plot[f[k,x],{x,0,1},PlotRange->{0,0.3}],
{k,15}]
Out[6]=
-Graphics-

```

Selecciona uno de los gráficos con el botón izquierdo del ratón y haz doble click para ver la animación. Si va muy rápida puedes controlar la velocidad con los botones que habrán aparecido en la esquina inferior izquierda de la ventana.

Ejemplo 2

Definamos $f_n: \mathbb{R}^+ \rightarrow \mathbb{R}$ por $f_n(x) = n(\sqrt[n]{x} - 1)$. Sin duda recuerdas que $\lim\{f_n(x)\} = \log x$, para todo $x \in \mathbb{R}^+$. Las gráficas y la animación que siguen, en las que he introducido pequeñas variantes que te permiten elegir diversos parámetros, te ayudarán a visualizar la convergencia de esta sucesión.

```

In[7]:=
Clear["Global`*"]

In[8]:=
g[n_,x_]:= n(x^(1/n)-1)

In[9]:=
grafgk[n_,a_,b_,c_,d_]:=
Plot[Evaluate[Table[g[k,x], {k,n}]],
{x,a,b}, PlotRange->{c,d},
PlotStyle->Table[RGBColor[0,0,1],{n}],
DisplayFunction->Identity]

In[10]:=
graflog[a_,b_,c_,d_]:= Plot[Log[x],{x,a,b},
PlotRange->{c,d}, PlotStyle->RGBColor[1,0,0],
DisplayFunction->Identity]

In[11]:=
graficas[n_,a_,b_,c_,d_]:=
Show[grafgk[n,a,b,c,d], graflog[a,b,c,d],
DisplayFunction->${DisplayFunction}]

In[12]:=
grafgklog[n_,a_,b_,c_,d_]:= Plot[{g[n,x],Log[x]},
{x,a,b},PlotRange->{c,d},
PlotStyle->{RGBColor[0,0,1],RGBColor[1,0,0]}]

In[13]:=
animacion[n_,a_,b_,c_,d_]:=
Table[grafgklog[k,a,b,c,d],{k,n}]

```

Escribe ahora `graficas[10,0.3,15,-1,5.5]`. Cambia algunos parámetros para ver qué pasa. Observa cómo la aproximación es peor al aumentar el intervalo $[a,b]$. Finalmente, para ver la animación puedes escribir `animacion[15,0.3,55,-1,6]`. Experimenta un poco al cambiar los parámetros.

Como hemos visto en clase, la sucesión que estamos considerando converge uniformemente en intervalos cerrados y acotados contenidos en \mathbb{R}^+ pero no converge uniformemente en ningún intervalo del tipo $]0, a[$ (porque $f_n(e^{-n}) - \log(e^{-n}) = n(1/e - 1) + n = n/e$) ni tampoco en ningún intervalo del tipo $[a, +\infty[$ (porque $f_n(e^n) - \log(e^n) = n(e - 1) - n = (e - 2)n$).

Series de Taylor

Dada una sucesión de funciones $\{f_n\}$, podemos formar otra, $\{F_n\}$, cuyos términos se obtienen sumando consecutivamente los de $\{f_n\}$. Es decir, $F_1 = f_1$, $F_2 = f_1 + f_2$, $F_3 = f_1 + f_2 + f_3$, en general $F_n = \sum_{k=1}^n f_k$.

La sucesión $\{F_n\}$ así definida se llama *serie de término general* f_n y la representaremos por el símbolo $\sum_{n \geq 1} f_n$. Los conceptos de convergencia puntual y uniforme para sucesiones de funciones se aplican igual para series. Así el campo de convergencia puntual de la serie $\sum_{n \geq 1} f_n$ cuyas funciones f_n suponemos

definidas en un intervalo I , es el conjunto $C = \{x \in I : \sum_{n \geq 1} f_n(x) \text{ es convergente}\}$. La única novedad es que ahora también podemos considerar el *campo de convergencia absoluta* de la serie, que es el conjunto $A = \{x \in I : \sum_{n \geq 1} |f_n(x)| \text{ es convergente}\}$.

Unas series de funciones muy importantes son las llamadas "series de potencias" que son series del tipo $\sum_{n \geq 0} c_n(x-a)^n$ donde $\{c_n\}_{n \geq 0}$ es una sucesión de números reales llamados *coeficientes de la serie*, y a es un número real. Se dice que la serie $\sum_{n \geq 0} c_n(x-a)^n$ *está centrada en* a . Las series de potencias más

frecuentes y útiles son las *series de Taylor*. Dada una función f que tiene derivadas de todos órdenes en un punto a , se define la serie de Taylor de f en el punto a como la serie de potencias $\sum_{k \geq 0} \frac{f^{(k)}(a)}{k!} (x-a)^k$, con los convenios usuales. Recuerda que la suma parcial n -ésima de la serie de Taylor es el polinomio de Taylor de orden n de f en a , que representamos como $T_n(f, a)(x) = f(a) + \sum_{k=1}^n \frac{f^{(k)}(a)}{k!} (x-a)^k$. Observa que la serie de Taylor es la sucesión de los polinomios de Taylor.

Vamos a realizar una animación mostrando cómo los gráficos de los polinomios de Taylor del coseno se van aproximando a la gráfica del coseno.

El comando `Series[func, {x, a, grado}]` calcula la serie de Taylor de `func` en la variable `x` centrada en `a` hasta los términos que indica `grado`. Escribe `Series[Cos[x], {x, 0, 10}]` y observa que al final de la respuesta al comando aparece el *orden de error* $O[x]^{11}$, que nos indica que los términos que faltan de la serie son todos de grado mayor o igual que once. Fíjate que en el caso del coseno, los términos de potencia 11 no existen pues, al ser una función par, las derivadas de orden impar del coseno en cero son nulas, pero el comando usado no informa a *Mathematica* de este detalle. Para obtener solamente el polinomio de Taylor sin que aparezca el término de error basta escribir `Normal[Series[Cos[x], {x, 0, 10}]]`.

Vamos a definir una función que calcule polinomios de Taylor en cero del coseno. Para ello escribe `Table[scos[n, x_] = Normal[Series[Cos[x], {x, 0, n}]], {n, 0, 100, 2}];` no olvides el `;` al final para evitar que la respuesta salga en pantalla y la ocupe toda. El comando anterior crea una lista con los polinomios de Taylor hasta el orden 100 y les da el nombre `scos[n, x]`. Como los polinomios de grados $2n$ y $2n+1$ coinciden, hemos hecho que el *contador* `n` tome solo los valores pares desde cero a cien. Puedes comprobarlo pidiendo a *Mathematica* `scos[n, x]` para distintos valores

pares de n .

Definamos ahora una función que realice pero no muestre en pantalla el gráfico del coseno junto con el de $\text{scos}[n, x]$ en un intervalo $[a, b]$.

```
In[14]:=
graf[n_, a_, b_] := Plot[ {Cos[x], scos[n, x]}, {x, a, b},
PlotRange -> {-2, 2},
PlotStyle -> {RGBColor[1, 0, 0], RGBColor[0, 0, 1]}]
```

Escribe `graf[10, -2Pi, 2Pi]` para ver la gráfica del coseno y de su polinomio de Taylor de orden 10, en $[-2\pi, 2\pi]$. Puedes variar el intervalo para comprobar que al aumentarlo la aproximación en puntos lejanos es mala.

Finalmente, como ya vimos antes, para hacer una animación cualquiera es necesario hacer todos los cuadros de la animación, como en un dibujo animado. Nuestra animación constará de 21 cuadros que serán los valores de `graf[n, a, b]` para n tomando valores de 0 a 40 de 2 en 2. Esto ya sabrás cómo hacerlo, ¿verdad? Es fácil, basta escribir `animacos[a_, b_] := Table[graf[n, a, b], {n, 0, 20, 2}]` y *Mathematica* se encarga del resto. Compruébalo escribiendo `animacos[-2Pi, 2Pi]`. Una vez que tienes los cuadros de la animación, seleccionas uno de ellos y haces doble click. Puedes controlar la velocidad como en los casos anteriores.

Ejercicios

Experimenta con lo que has aprendido, modificando de forma conveniente los comandos, para realizar una animación de la convergencia de la series de Taylor del seno ($\text{Sin}[x]$) y del arcotangente ($\text{ArcTan}[x]$) en $x = 0$. Elige intervalos y rango de representación adecuados. ¿Observas algún comportamiento particular en las aproximaciones al arcotangente? ¿Sabes explicarlo?